# Maintainer: Lukas Fleischer <lfleischer@archlinux.org>
# Contributor: noonov <noonov@gmail.com>

pkgname=zshdb
pkgver=1.1.3
pkgrel=1
pkgdesc='A debugger for zsh scripts'
arch=('any')
url='https://github.com/rocky/zshdb/'
license=('GPL')
depends=('zsh' 'coreutils')
optdepends=('python-pygments: for syntax highlighting')
source=("$pkgname-$pkgver.tar.gz::https://github.com/rocky/$pkgname/archive/$pkgver.tar.gz")
sha512sums=('64d08fc890b503136c8b45dea528d56a5a584cc91fa723fe0785be8c2953395f860b58d363465cc77057cecc579c2ca4ad3372d73106b65ac9220c2adbc680ae')
b2sums=('3dfab5677fb173e8a16508f0865661d12eca550dfc8b60bc1b107f0b92c821e837912308c3b3bb398575e78ca54fa4a33844769ef6d758700e2c28d1036c2b55')

build() {
  cd "$pkgname-$pkgver"

  ./autogen.sh
  ./configure --prefix=/usr
  make
}

package() {
  cd "$pkgname-$pkgver"

  make DESTDIR="$pkgdir" install
}
